
Executable Specifications

Given the similarity between natural deduction,

Inference Step

premise1 · · · premisen
(condition)

conclusion

Assumption

conclusion

and Horn clause logic,

Inference Step P0 :- P1, . . . ,Pn.

Fact P0 :- true.

it is natural to assume that we can implement our operational semantics
in Prolog.



New Syntax

In order to do this we have to rewrite our syntax specification in terms of Prolog terms.

Arithmetic expressions:

A ::= n | x | add(A,A) | sub(A,A) |mult(A,A)

where n is any valid Prolog integer value and x any valid Prolog object name.
Boolean expressions:

B ::= true | false | eq(A,A) | le(A,A) | not(B) | and(B,B) | or(B,B)

Commands:

C ::= skip | assign(x ,A) | seq(C ,C) | if(B,C ,C) |whiledo(B,C)

Note: the seq operator is made infix for convenience - just remember that it is
left-associative - under certain circumstances you will need to help the Prolog parser
out with parentheses around the appropriate terms.



Programs

Example 1:
v := 1; if v ≤ 0 then v := (−1) ∗ v else skip end

assign(v, 1) seq if(le(v, 0), assign(v,mult(−1, v)), skip)

Example 2:

n := 5; y := 1; while 2 ≤ n do (y := n ∗ y ; n := n − 1) end

assign(n, 5) seq assign(y, 1) seq whiledo(le(2, n), assign(y,mult(n, y)) seq assign(n, sub(n, 1)))



State

Prolog does not allow us to pass functions around (it is a first-order
language), therefore, we cannot use the definition of state from our
natural deduction operational semantics.

However, consider the following,

σ[m/x ][n/y ][k/z ],

This could be interpreted as a list of variable bindings applied to the
state σ if we interpret the [. . .] as list constructors and juxtaposition as a
list append operation,

σ[m/x , n/y , k/z ].

In Prolog we can model this as the term structure

state([bind(k , z),bind(n, y),bind(m, x)], s)

where k, n,m ∈ I and x , y , z ∈ Loc and s represents an arbitrary state.



State

If the state is the initial state, where

σ0[m/x , n/y , k/z ]

then
state([bind(k , z),bind(n, y),bind(m, x)], s0)

where s0 is a reserved symbol for the initial state in our representation.

Given our representation we have an interesting equivalence, given a
state s, then

s ≡ state([ ], s)

We will make use of this equivalence when encoding our semantic
predicates.



State

Since we turned the state representation from a function into a list, we
now have to adjust the variable lookup mechanism. We do this via the
predicate lookup,

% the predicate ’lookup(+Variable,+State,-Value)’ looks up

% the variable in the state and returns its bound value.

:- dynamic lookup/3. % modifiable predicate

lookup(_,s0,0).

lookup(X,state([],S),Val) :-

lookup(X,S,Val).

lookup(X,state([bind(Val,X)|_],_),Val).

lookup(X,state([_|Rest],S),Val) :-

lookup(X,state(Rest,S),Val).



Arithmetic Expression Summary

Recall our natural deduction definition for arithmetic expressions:

for n ∈ D and σ ∈ Σ
(n, σ) 7→ eval(n)

for x ∈ Loc and σ ∈ Σ
(x , σ) 7→ σ(x)

(a0, σ) 7→ k0 (a1, σ) 7→ k1
where k = k0 + k1

(a0 + a1, σ) 7→ k

(a0, σ) 7→ k0 (a1, σ) 7→ k1
where k = k0 − k1

(a0 − a1, σ) 7→ k

(a0, σ) 7→ k0 (a1, σ) 7→ k1
where k = k0 × k1

(a0 ∗ a1, σ) 7→ k

NOTE: k , k0, k1 ∈ I, a0, a1 ∈ Aexp, and σ ∈ Σ.



Prolog Aexp Semantics

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

% semantics of arithmetic expressions

(C,_) -->> C :- % constants

int(C),!.

(X,State) -->> Val :- % variables

atom(X),

lookup(X,State,Val),!.

(add(A,B),State) -->> Val :- % addition

(A,State) -->> ValA,

(B,State) -->> ValB,

Val xis ValA + ValB,!.

(sub(A,B),State) -->> Val :- % subtraction

(A,State) -->> ValA,

(B,State) -->> ValB,

Val xis ValA - ValB,!.

(mult(A,B),State) -->> Val :- % multiplication

(A,State) -->> ValA,

(B,State) -->> ValB,

Val xis ValA * ValB,!.

Note: The cut predicate is necessary to make sure that these rules are interpreted as state transitions, i.e., once a

state transition has occurred in an abstract machine it cannot be undone.



The -->> Predicate

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

% the predicate ’(+Syntax,+State) -->> -SemanticValue’ computes

% the semantic value for each syntactic structure

:- op(700,xfx,-->>).

:- dynamic (-->>)/2. % modifiable predicate

:- multifile (-->>)/2.



The ‘xis’ Predicate

bash-3.2$ prolog -f prolog-semantics/preamble.pl

% xis.pl compiled 0.00 sec, 6,960 bytes

% /Users/lutz/Documents/csc501/prolog-semantics/preamble.pl compiled 0.00 sec, 9,532 bytes

Welcome to SWI-Prolog (Multi-threaded, 32 bits, Version 5.10.1)

Copyright (c) 1990-2010 University of Amsterdam, VU Amsterdam

SWI-Prolog comes with ABSOLUTELY NO WARRANTY. This is free software,

and you are welcome to redistribute it under certain conditions.

Please visit http://www.swi-prolog.org for details.

For help, use ?- help(Topic). or ?- apropos(Word).

?- X is 1 + 2.

X = 3.

?- X xis 1 + 2.

X = 3.

?- X is y + 2.

ERROR: is/2: Arithmetic: ‘y/0’ is not a function

?- X xis y + 2.

X = y+2.

?-

Note: both the -->> and the xis predicate are defined in the preamble.pl file.

Suggestion: put the preamble.pl file in a known place such as the super-directory of all

your projects, then you can load it as ’prolog -f ../preamble.pl’.



Evaluation of Arithmetic Expressions

Let ae ≡ (2 ∗ 3) + 5, prove that the semantic value of this expression in
some state s is equal to 11 using the Prolog semantics (assume that the
semantics is given in the file ’sem.pl’).

?- [’sem.pl’].

% preamble.pl compiled 0.00 sec, 900 bytes

% xis.pl compiled 0.00 sec, 6,788 bytes

% sem.pl compiled 0.00 sec, 14,164 bytes

true.

?- (add(mult(2,3),5),s) -->> V, V = 11.

V = 11.

?-



Evaluation of Arithmetic Expressions

Now, let ae ≡ x + 1, where x ∈ Loc, prove that the semantic value of
this expression in some state s is equal to vx + 1 where lookup(x,s,vx).

?- [’sem.pl’].

% preamble.pl compiled 0.00 sec, 900 bytes

% xis.pl compiled 0.00 sec, 6,788 bytes

% sem.pl compiled 0.00 sec, 14,164 bytes

true.

?- asserta(lookup(x,s,vx)).

true.

?- (add(x,1),s) -->> vx+1.

true.

?-

Note: the predicate ’asserta’ is preferable because it inserts the clause at

the top of the rule database.



Evaluation of Arithmetic Expressions

Now, let ae ≡ x + 1, where x ∈ Loc, prove that the semantic value of
this expression in the initial state s0 equal to 1.

?- [’sem.pl’].

% preamble.pl compiled 0.00 sec, 900 bytes

% xis.pl compiled 0.00 sec, 6,788 bytes

% sem.pl compiled 0.00 sec, 14,164 bytes

true.

?- (add(x,1),s0) -->> 1.

true.

?-



Evaluation of Arithmetic Expressions

Consider the following proof: Let ae ≡ x + 3 ∗ 5, where x ∈ Loc, prove that the
semantic value of this expression in some state s is equal to vx + 15 where
lookup(x,s,vx).

?- [’sem.pl’].

% preamble.pl compiled 0.00 sec, 900 bytes

% xis.pl compiled 0.00 sec, 6,788 bytes

% sem.pl compiled 0.00 sec, 14,164 bytes

true.

?- asserta(lookup(x,s,vx)).

true.

?- (add(x,mult(3,5)),s) -->> vx+15.

true.

?-



Theorem Proving with Prolog

Using Prolog as a theorem prover:

1 The Prolog Meta-Language – we can consult programs, assert assumptions,
retract assumptions, and query a program in order to prove a theorem.

2 Universally Quantified Variables in Queries – consider the proof,

?- (mult(3,5),s) -->> V, V = 15.

can be interpreted in standard FOL as,

∀s∃V [(mult(3, 5), s)−−� V ∧=(V , 15)].

⇒ We use symbolic constants in queries to express universally quantified
variables.

3 Proof Scores – we can write a proof as a Prolog meta-language program.

4 Soundness – under certain circumstances the default resolution rule can be
unsound, to avoid this insert the following code into your proof scores:

:- set_prolog_flag(occurs_check,true).

If you use the ’preamble.pl’ file this is done automatically for you.



Theorem Proving with Prolog

The second point on the previous slide deserves some additional
attention. Consider for the moment that we would like to prove

∀s[(mult(3, 5), s)−−� 15].

If we write this blindly as a query using standard Prolog variables, then

?- (mult(3,5),S) -->> 15.

Now interpreting this query according to Prolog, then

∃s[(mult(3, 5), s)−−� 15].

That means, this query does not prove our intended proof goal.



Theorem Proving with Prolog

From FOL quantification theory we have the following axiom (Universal
Generalization). Let p be a predicate and let y ∈ U be some arbitrarily chosen
element of some universe U, then

p(y)

∴ ∀x[p(x)]

with x ∈ U. In plain English,

If I can show that a predicate holds for an arbitrarily chosen element of
some universe, then I can infer that this predicate holds for all elements of
that universe.

With this we can rewrite our query as

?- (mult(3,5),y) --> 15.

Here the lowercase y is an element of some universe, in this case the States, and
therefore, if Prolog can prove this goal, we can conclude that

∀x[(mult(3, 5), x)−−� 15].

with x ∈ States.



Theorem Proving with Prolog

We have to be careful with universal generalization; the statement “some arbitrarily
chosen element of some universe” has a specific meaning:

The element is not allowed to reveal its structure or internal state.

Consequently, the predicate we want to generalize is not allowed to investigate the
structure or state of the element.



Theorem Proving with Prolog

Example: Let Σ be the set of all states and let σ′ ∈ Σ be some arbitrarily chosen
element of that set. Let p(σ′) ≡ σ′(x) = 20. Now, applying universal generalization
we have,

p(σ′)

∴ ∀σ[p(σ)]

with σ ∈ Σ. This is clearly a fallacious argument, there will be many states in which
σ(x) 6= 20. This argument failed because the predicate p investigated the internal
structure of σ′.

Example: Let Σ be the set of all states and let σ′ ∈ Σ be some arbitrarily chosen
element of that set. Let p(σ′) ≡ σ′[20/x](x) = 20. Now, applying universal
generalization we have,

p(σ′)

∴ ∀σ[p(σ)]

with σ ∈ Σ. This argument clearly holds because we did not look at the internal
structure of the element.



Theorem Proving with Prolog

Example: Let N be the set of all natural numbers and let 7 ∈ N be some arbitrarily
chosen element of that set. Let p(7) ≡ 7 ≤ 100. Now, applying universal
generalization we have,

p(7)

∴ ∀k[p(k)]

with k ∈ N. Again, this argument fails because we allowed the predicate to investigate
the structure of the element (the value 7).

Example: Let N be the set of all natural numbers and let n ∈ N be some arbitrarily
chosen element of that set. Let p(n) ≡ n ≤ n + 100. Now, applying universal
generalization we have,

p(n)

∴ ∀k[p(k)]

with k ∈ N. This argument succeeds because we did not look at the structure (specific
value) of the element.



Proof Scores

% load preamble

:- [’preamble.pl’].

% proof1.pl

% Proof score:

%

:- >>> ’Show that’.

:- >>> ’(forall x)(forall s)(forall vx)(exists V)[(add(x,mult(3,5)),s)-->>V ^ =(V,vx+15)]’.

:- >>> ’ assuming lookup(x,s,vx)’.

% load semantics

:- [’sem.pl’].

% state our assumption

:- asserta(lookup(x,s,vx)).

% run the proof

:- (add(x,mult(3,5)),s)-->>V, V = vx + 15.



Expression Equivalence

In our Prolog framework semantic equivalence between arithmetic
expression can be formulated as follows:

a0 ∼ a1 iff ∀s,∃V0,V1 [(a0, s)−−� V0 ∧ (a1, s)−−� V1 ∧=(V0,V1)] ,

for a0, a1 ∈ Aexp.



Expression Equivalence

% load preamble

:- [’preamble.pl’].

% proof-equiv.pl

:- >>> ’ prove that mult(2,3) ~ add(3,3)’.

%

% show that

% (forall s)(exists V0,V1)

% [(mult(2,3),s)-->>V0 ^ (add(3,3),s)-->>V1 ^ =(V0,V1)]

% load semantics

:- [’sem.pl’].

% proof

:- (mult(2,3),s)-->>V0 , (add(3,3),s)-->>V1 , V0 = V1.



Expression Equivalence

?- [’proof-equiv.pl’].

% preamble.pl compiled 0.00 sec, 924 bytes

>>> prove that mult(2,3) ~ add(3,3)

% preamble.pl compiled 0.00 sec, 128 bytes

% xis.pl compiled 0.00 sec, 6,788 bytes

% sem.pl compiled 0.00 sec, 12,548 bytes

% proof-equiv.pl compiled 0.01 sec, 14,876 bytes

true.

?-



Expression Equivalence

% proof-comm.pl

:- [’preamble.pl’].

:- >>> ’prove that add(a0,a1) ~ add(a1,a0)’.

%

% show that

% (forall s,a0,a1)(exists V0,V1)

% [sem(add(a0,a1),s,V0)^sem(add(a1,a0),s,V1)^=(V0,V1)]

% assuming

% (a0,s) -->> va0.

% (a1,s) -->> va1.

% load semantics

:- [’sem.pl’].

% assumptions on semantic values of expressions

:- asserta((a0,s)-->>va0).

:- asserta((a1,s)-->>va1).

% assumption on integer addition commutativity

:- asserta(comm(A + B, B + A)).

% proof

:- (add(a0,a1),s)-->>V0, (add(a1,a0),s)-->>V1,comm(V0,VC0),VC0=V1.


